


EXPERT SYSTEM CASE STUDY #1

"SELECTING A TOY FOR A CHILD"

CS 375 AND FYS AND CS 434
Artificial Intelligence

Dr. J. Wunderlich



- DOUBLE CLICK ON VPX.EXE
- PRESS "4" TO CONSULT
- HIT ENTER TO PICK "~~EXP~~ FZ"
- PRESS "2" TO EXECUTE

• HIT ARROWS TO MOVE AROUND SELECTIONS, HIT ENTER TO TAG CERTAIN ONES

HIT "HOME" TO ENTER CONFIDENCE VALUES

DO "5" TO SHOW OTHER RULES

TO SET TRACE, DO "6" TO SET, THEN "2" FOR

THEN "6" TO QUIT, "2" TO GO STATUS TRACE
USE TREE (TEXT OR GRAPHICS)

TO SHOW TREE, DO "5" FOR TREE, THEN "3" FOR

#2) Develop an expert system to help a customer decide on a toy for a young child. (VP-EXPERT was used)

ASSUMPTIONS:

- 1) Gender is not relevant for children younger than the age of one.
- 2) Children younger than the age of one cannot have a known preference for toys.
- 3) Children between the ages of one and three cannot have a known preference for toys.

CONTROL OF SEARCH SPACE:

As a result of the above assumptions, the rules have been ordered so that:

- A) child_age is the first variable in the premise of every rule.
- B) gender and child_preference questions are not asked when child_age = under_1
- C) The child_preference question is not asked when child_age = one_to_three

TESTING OF EXPERT SYSTEM:

The expert system was tested under the three scenarios listed in the table below.

The resulting output is shown in the last column with the calculated confidence factors.

These confidence factors were calculated by VP-EXPERT using the following standard laws of Certainty:

- A) $CNF(P1 \text{ and } P2) = \min(CNF(P1), CNF(P2))$
- B) $CNF(P1 \text{ or } P2) = \max(CNF(P1), CNF(P2))$

And when two or more rules support the same result R:

- C) $CNF(R1) + CNF(R2) - (CNF(R1) * CNF(R2))$ when $CNF(R1)$ and $CNF(R2)$ are positive
- D) $CNF(R1) + CNF(R2) + (CNF(R1) * CNF(R2))$ when $CNF(R1)$ and $CNF(R2)$ are negative
- E) $(CNF(R1) + CNF(R2)) / (1 - \min(|CNF(R1)|, |CNF(R2)|))$ otherwise

Confidence factors accepted by VP-EXPERT are (0 to 100) unlike the standard (-1 to 1); One solution is scale by 1/100:
 The example below shows the CNF calculations for suggestedd_toy = dress_up_doll

For the premise of RULE 12:

$[(\text{child_age}=\text{four_to_six}) \text{ AND } (\text{price}=\text{under_25}) \text{ AND } (\text{gender}=\text{female}) \text{ AND } (\text{child_preference}=\text{cuddly_toy})]$
 $CNF=1 \quad \text{AND} \quad CNF=0.65 \quad \text{AND} \quad CNF=1 \quad \text{AND} \quad CNF=0.55$

Using law #A above; $CNF(\text{premise}) = \min(1, 0.65, 1, 0.55) = 0.55$

Using law #C above; $CNF(\text{conclusion}) = CNF(\text{premise}) * CNF(\text{RULE 12}) = [0.55 * 0.9] = 0.49$ see NOTE 7

Laws #C, #D, or #E are not used here because only RULE 12 fires to support the goal (suggested_toy=dress_up_doll)

TEST TRACE #	INPUT				RULE FIRED AND ITS CNF (see NOTE 7)	OUTPUT suggested_toy
	child_age (see NOTE 1)	price (see NOTE 2)	gender (see NOTE 3)	child_preference (see NOTE 4)		
1	under_1	under_25 (CNF=65) over_25 (CNF=20)	N.A.	N.A.	CNF(R1)=95 CNF(R2)=90 CNF(R3)=95 CNF(R4)=75	teething_toy (CNF=61) mobile_for_crib (CNF=58) plastic_rattle (CNF=61) sterling_silver_rattle (CNF=15)
2	one_to_three	under_25 (CNF=65) over_25 (CNF=20)	male	N.A.	CNF(R5)=90 CNF(R6)=90 CNF(R7)=85	roly_poly (CNF=58) tricycle (CNF=18) hammer_and_pegs_game (CNF=55)
3	four_to_six <i>CNF=100</i>	under_25 (CNF=65) over_25 (CNF=20)	female <i>CNF=100</i>	action_toys (CNF=25) cuddly_toys (CNF=55) creative_toys (CNF=75)	CNF(R9)=95 CNF(R11)=95 CNF(R12)=90 CNF(R14)=85	lincoln_logs (CNF=61) doll_house (CNF=18) <u>dress_up_doll (CNF=49)</u> toy_tea_set (CNF=55)

NOTE 1: Question to user is: "What is child's age? (ONLY SELECT ONE)" ---CNF defaults to 100---

NOTE 2: Question to user is: "How much do you want to spend? (ASSIGN CONFIDENCE VALUES TO EACH)"

NOTE 3: Question to user is: "What is child's gender? (ONLY SELECT ONE)" ---CNF defaults to 100---

NOTE 4: Question to user is: "What kind of toys does the child like? (ASSIGN CONFIDENCE VALUES TO EACH)"

NOTE 5: The rules of the expert system are attached on the following pages.

NOTE 6: The test traces are attached on the following pages after the rules.

NOTE 7: The CNF's for RULES assigned by the knowledge engineer after consultation with domain expert.

Dr. W.

Also Dr. W. HERE (BUT NO REPLY FOR WHAT THE IDEA! AN INTERVIEW AN EXPERT!

ACTIONS

FIND suggested_toy
DISPLAY "Suggested toys:
{#suggested_toy}";

RULE 1

IF child_age = under_one AND
price = under_25
THEN suggested_toy = teething_toy CNF 95;

RULE 2

IF child_age = under_one AND
price = under_25
THEN suggested_toy = mobile_for_crib CNF 90;

RULE 3

IF child_age = under_one AND
price = under_25
THEN suggested_toy = plastic_rattle CNF 95;

RULE 4

IF child_age = under_one AND
price = over_25
THEN suggested_toy = sterling_silver_rattle CNF 75;

RULE 5

IF child_age = one_to_three AND
price = under_25
THEN suggested_toy = roly_poly CNF 90;

RULE 6

IF child_age = one_to_three AND
price = over_25
THEN suggested_toy = tricycle CNF 90;

RULE 7

IF child_age = one_to_three AND
price = under_25 AND
gender = male
THEN suggested_toy = hammer_and_pegs_game CNF 85;

RULE 8

IF child_age = one_to_three AND
price = over_25 AND
gender = female
THEN suggested_toy = small_rocking_chair CNF 80;

RULE 9

IF child_age = four_to_six AND
price = under_25 AND
child_preference = creative_toys
THEN suggested_toy = lincoln_logs CNF 95;

RULE 10

IF child_age = four_to_six AND
price = over_25 AND
gender = male AND
child_preference = action_toys
THEN suggested_toy = go_cart CNF 85;

```

RULE 11
IF child_age = four_to_six AND
price = over_25 AND
gender = female AND
child_preference = creative_toys
THEN suggested_toy = doll_house CNF 90;

RULE 12
IF child_age = four_to_six AND
price = under_25 AND
gender = female AND
child_preference = cuddly_toys
THEN suggested_toy = dress_up_doll CNF 90;

RULE 13
IF child_age = four_to_six AND
price = over_25 AND
gender = male AND
child_preference = action_toys
THEN suggested_toy = hot_wheels_set CNF 95;

RULE 14
IF child_age = four_to_six AND
price = under_25 AND
gender = female AND
child_preference = creative_toys
THEN suggested_toy = toy_tea_set CNF 85;

RULE 15
IF child_age = four_to_six AND
price = under_25 AND
gender = male AND
child_preference = creative_toys
THEN suggested_toy = army_men CNF 90;

```

```

ASK child_age : "What is child's age? (ONLY SELECT ONE)" ;
CHOICES child_age : under_one, one_to_three, four_to_six ;

```

```

ASK price : "How much do you want to spend? (ASSIGN CONFIDENCE VALUES TO EACH)";
CHOICES price : under_25, over_25;

```

```

ASK gender : "What is the child's gender? (ONLY SELECT ONE)";
CHOICES gender : male, female;

```

```

ASK child_preference : "what kind of toys does the child like?
                        (ASSIGN CONFIDENCE VALUES TO EACH)";
CHOICES child_preference : action_toys, cuddly_toys, creative_toys;

```

```

PLURAL : suggested_toy, price, child_preference;

```

TRACE #1

```
Testing b:\F2.kbs
(= yes CNF 0 )
!   suggested_toy
!   !   Testing 1
!   !   !   child_age
!   !   !   !   (= under_one CNF 100 )
!   !   !   !   price
!   !   !   !   !   (= under_25 CNF 65 )
!   !   !   !   !   !   (= over_25 CNF 20 )
!   !   !   (= teething_toy CNF 61 )
!   !   Testing 2
!   !   (= mobile_for_crib CNF 58 )
!   !   Testing 3
!   !   (= plastic_rattle CNF 61 )
!   !   Testing 4
!   !   (= sterling_silver_rattle CNF 15 )
!   !   Testing 5
!   !   Testing 6
!   !   Testing 7
!   !   Testing 8
!   !   Testing 9
!   !   Testing 10
!   !   Testing 11
!   !   Testing 12
!   !   Testing 13
!   !   Testing 14
!   !   Testing 15
```

TRACE #2

```
Testing B:\F2.kbs
(= yes CNF 0 )
!   suggested_toy
!   !   Testing 1
!   !   !   child_age
!   !   !   !   (= one_to_three CNF 100 )
!   !   Testing 2
!   !   Testing 3
!   !   Testing 4
!   !   Testing 5
!   !   !   price
!   !   !   !   (= under_25 CNF 65 )
!   !   !   !   (= over_25 CNF 20 )
!   !   (= roly_poly CNF 58 )
!   !   Testing 6
!   !   (= tricycle CNF 18 )
!   !   Testing 7
!   !   !   gender
!   !   !   !   (= male CNF 100 )
!   !   (= hammer_and_pegs_game CNF 55 )
!   !   Testing 8
!   !   Testing 9
!   !   Testing 10
!   !   Testing 11
!   !   Testing 12
!   !   Testing 13
!   !   Testing 14
!   !   Testing 15
```

TRACE #3

```
Testing B:\F2.kbs
(= yes CNF 0 )
! suggested_toy
!   Testing 1
!     ! child_age
!     !   (= four_to_six CNF 100 )
!   Testing 2
!   Testing 3
!   Testing 4
!   Testing 5
!   Testing 6
!   Testing 7
!   Testing 8
!   Testing 9
!     price
!     !   (= under_25 CNF 65 )
!     !   (= over_25 CNF 20 )
!     child_preference
!     !   (= action_toys CNF 25 )
!     !   (= cuddly_toys CNF 55 )
!     !   (= creative_toys CNF 75 )
!   (= lincoln_logs CNF 61 )
! Testing 10
!   gender
!   !   (= male CNF 100 )
!   (= go_cart CNF 17 )
! Testing 11
! Testing 12
! Testing 13
!   (= hot_wheels_set CNF 19 )
! Testing 14
! Testing 15
!   (= army_men CNF 58 )
```